'    Copyright 2008 Jan Molnar <jan.molnar@abds.de>
'
'    This file is part of JCQ.

'    JCQ is free software: you can redistribute it and/or modify
'    it under the terms of the GNU General Public License as published by
'    the Free Software Foundation, either version 2 of the License, or
'    (at your option) any later version.

'    JCQ is distributed in the hope that it will be useful,
'    but WITHOUT ANY WARRANTY; without even the implied warranty of
'    MERCHANTABILITY or FITNESS FOR A PARTICULAR PURPOSE.  See the
'    GNU General Public License for more details.

'    You should have received a copy of the GNU General Public License
'    along with JCQ. If not, see <http://www.gnu.org/licenses/>.

Public Class WindowExtenderProvider
    Private Shared _lock As Object = New Object()
    Private Shared _ResizeExtenders As New Dictionary(Of Window, WindowResizeExtender)

    Public Shared Function AttachResizeExtender(ByVal wnd As Window) As WindowResizeExtender
        Dim extender As WindowResizeExtender

        extender = New WindowResizeExtender
        extender.Window = wnd

        SyncLock _lock
            _ResizeExtenders.Add(wnd, extender)
        End SyncLock

        AddHandler wnd.Closed, AddressOf OnWindowClosed

        Return extender
    End Function

    Private Shared Sub OnWindowClosed(ByVal sender As Object, ByVal e As EventArgs)
        Dim wnd As Window

        wnd = DirectCast(sender, Window)

        SyncLock _lock
            If _ResizeExtenders.ContainsKey(wnd) Then
                _ResizeExtenders.Remove(wnd)
            End If
        End SyncLock
    End Sub

    Public Shared Function GetResizeExtender(ByVal wnd As Window) As WindowResizeExtender
        SyncLock _lock
            If _ResizeExtenders.ContainsKey(wnd) Then
                Return _ResizeExtenders(wnd)
            Else
                Return Nothing
            End If
        End SyncLock
    End Function
End Class